/*
 * 光照传感器
 * 命令示例：
{
  "method": "get",
  "obj": ["illu"]
}
*/
#include "Arduino.h"    
#include <OneButton.h>       
#include <WiFi.h>
#include <SDDC_SDK_lib.h>
#include <cJSON.h>
#include <Wire.h>

// BH1750FVI in ADDR 'L' mode 7bit addr
#define ADDR 0b0100011

#define SDDC_CFG_PORT             680U             // SDDC 协议使用的端口号
#define PIN_INPUT                 0                // 选择 IO0 进行控制
#define ESP_TASK_STACK_SIZE       4096
#define ESP_TASK_PRIO             25

static const char* ssid     = "EOS-Tenda";         // WiFi 名
static const char* password = "1234567890";        // WiFi 密码
static double illu_val;
OneButton button(PIN_INPUT, true);

static void illu_sensor_task(void *arg)
{  
    double val = 0;
    int i;
    while(1)
    {
        // 获取光照数据
        Wire.beginTransmission(ADDR);
        Wire.write(0b00000111);
        if (0 != Wire.endTransmission()){
             Serial.println("endTransmission error \n");
        }
     
        Wire.beginTransmission(ADDR);
        Wire.write(0b00100000);
        if (0 != Wire.endTransmission()){
             Serial.println("endTransmission error \n");
        }
        // typical read delay 120ms
        delay(120);
        Wire.requestFrom(ADDR, 2); // 2byte every time
        for (i = 0; Wire.available() >= 1; ) 
        {
            char c = Wire.read();
            i = (i << 8) + (c & 0xFF);
        }
        val = i / 1.2;  
        illu_val = val;
        delay(5000);
    }
}
/* 
 *  单次获取数据
 */
sddc_bool_t single_get_sensor(char *objvalue, int value_len)
{
    double value = illu_val;
    snprintf(objvalue, value_len, "%lf", value);
    return SDDC_TRUE;
}
 
/* 
 *  数字量设备对象函数与处理方法注册
 */
NUM_DEV_REGINFO num_dev[] = {
  
};

/*
 *  显示设备对象函数与处理方法注册
 */
DIS_DEV_REGINFO dis_dev[] = {

};

/*
 * IO设备对象设置函数与处理方法注册
 */
IO_DEV_REGINFO io_dev[] = {

};

/*
 *  系统对象状态获取注册
 */
DEV_STATE_GET  dev_state_get_reg[] = {
        {"illu",   DEV_NUM_TYPE,  single_get_sensor},

};

/*
 *  当前设备的信息定义
 */
DEV_INFO    dev_info = {
            .name     = "光照传感器",
            .type     = "device.illu",
            .excl     = SDDC_FALSE,
            .desc     = "ESP-32S + HB1750VI",
            .model    = "IDILLU01B",
            .vendor   = "inspiration-desktop",
};

/*
 *   系统注册对象汇聚
 */
SDDC_CONFIG_INFO sys_cfg = {
        .token             = "1234567890",            // 设备密码
        .devinfo           = &dev_info,               
        .io_dev_reg        = io_dev,
        .io_dev_reg_num    = ARRAY_SIZE(io_dev),
        .num_dev_reg       = num_dev,
        .num_dev_reg_num   = ARRAY_SIZE(num_dev),
        .state_get_reg     = dev_state_get_reg,
        .state_get_reg_num = ARRAY_SIZE(dev_state_get_reg),
        .dis_dev_reg       = dis_dev,
        .dis_dev_num       = ARRAY_SIZE(dis_dev),
};


/*
 * IO0 按键检测任务
 */
static void esp_io0_key_task()
{
    WiFi.disconnect();
    while (WiFi.status() != WL_CONNECTED)
    {
        delay(1000);
        Serial.println("WIFI Wait for Smartconfig");
          
        // 设置为Station模式
        WiFi.mode(WIFI_STA);
          
        // 开始智能配网
        WiFi.beginSmartConfig();
        while (1)
        {
            delay(500);
            // 检查智能配网
            if (WiFi.smartConfigDone())
            {
                // 设置自动连接
                WiFi.setAutoConnect(true);  
                break;
            }
        }
    }
}

/*
 * 循环扫描按键
 */
static void esp_tick_task(void *arg)
{
    void *sddc = arg;
    (void)sddc;
    while(1)
    {
        button.tick();
        delay(100);
    }
}

/*
 * 初始化传感器
 */
void sensor_init()
{
    // 初始化I2C
    Wire.begin();
    Wire.beginTransmission(ADDR);
    Wire.write(0b00000001);
    Wire.endTransmission();
    
    // 创建传感器任务，周期性传感器的数据并发送给 EdgerOS
    xTaskCreate(illu_sensor_task, "illu_sensor_task", ESP_TASK_STACK_SIZE, NULL, ESP_TASK_PRIO, NULL);
}

void setup() {
    byte mac[6];
    int i = 0;
    Serial.begin(115200);
    Serial.setDebugOutput(true);
    Serial.println();

    // 清除一下按键状态机的状态
    button.reset();
    
    // 创建按键扫描线程，长按 IO0 按键，松开后ESP32 将会进入 SmartConfig 模式
    sddc_printf("长按按键进入 Smartconfig...\n");
    button.attachLongPressStop(esp_io0_key_task);
    xTaskCreate(esp_tick_task, "button_tick", ESP_TASK_STACK_SIZE, NULL, ESP_TASK_PRIO, NULL);

    // 启动 WiFi 并且连接网络
    printf("WiFi.status() = %d\n", WiFi.status());
    WiFi.begin(ssid, password);

    while (WiFi.status() != WL_CONNECTED) 
    {
        delay(1000);
        sddc_printf("WiFi.status() = %d\n", WiFi.status());
        Serial.print(".");
        i++;
        if (i >= 20 ) {
            WiFi.begin(ssid, password);
            i = 0;
        }
    }
    printf("WiFi.status() = %d\n", WiFi.status());
    // 获取并打印 IP 地址
    Serial.println("");
    Serial.println("WiFi connected");
    Serial.print("'ip :");
    Serial.print(WiFi.localIP());
    Serial.println("' to connect"); 

    // 初始化传感器
    sensor_init();

    // sddc协议初始化
    sddc_lib_main(&sys_cfg);

    // 获取并打印网卡 mac 地址
    WiFi.macAddress(mac);
    sddc_printf("MAC addr: %02x:%02x:%02x:%02x:%02x:%02x\n",
              mac[5], mac[4], mac[3], mac[2], mac[1], mac[0]);
    // 使用网卡 mac 地址设置设备唯一标识 UID
    sddc_set_uid(G_sddc, mac);
}

void loop() {
    // 运行 SDDC 协议循环
    while (1) 
    {
        sddc_printf("SDDC running...\n");
        if (sddc_run(G_sddc) == -1) {
            delay(10000);
            printf("WiFi.status() = %d\n", WiFi.status());
            ESP.restart();
        }
        sddc_printf("SDDC quit!\n");
    }

    // 销毁 SDDC 协议
    sddc_destroy(G_sddc);
}
